
Abstract

Continuations are variously understood as representations of the
current evaluation context and as representations of the rest of the
computation, but these understandings contradict each other: plug-
ging an expression in a context yields a new expression whereas
sending an intermediate result to a continuation yields the final an-
swer. We show that continuations-as-evaluation-contexts are the
defunctionalized representation of the continuation of a single-
step reduction function and that continuations-as-the-rest-of-the-
computation are the continuation of an evaluation function. Fur-
thermore, we show that defunctionalizing the continuation of an
evaluator gives rise to the same evaluation contexts as in the single-
step reducer. The only difference is how these evaluation contexts
are interpreted: a ‘plug’ interpretation yields one-step reduction,
whereas a ‘refocus’ interpretation yields evaluation.

We then present a constructive corollary of Reynolds’s histor-
ical warning about depending on the evaluation order of a meta-
language for an interpreter: The two best-known abstract machines
for the λ-calculus, Krivine’s machine and Felleisen et al.’s CEK
machine, are in fact the call-by-name and call-by-value counter-
parts of thesame(evaluation-order dependent) interpreter for the
λ-calculus.

1 Introduction

The notion of continuation is ubiquitous in many different areas of
computer science, including logic, constructive mathematics, pro-
gramming languages, and programming. Nevertheless, continua-
tions are a remarkably elusive, even mystifying, notion. They pop
up virtually everywhere as a uniform solution to control-related
problems, and it seems that no two alternative solutions to these
problems are alike. Worse, no particular effort seems to have been
devoted to connecting these alternative solutions to the solutions
based on continuations and from there, to transpose these alterna-
tive solutions to other domains.

1.1 Continuations, informally

What is a continuation? To some, it is the representation of an eval-
uation context, i.e., an expression with a hole; plugging an expres-
sion into this hole yields a new expression. To others, a continuation
is a representation of the rest of the computation; sending it an in-
termediate result yields the final result of the overall computation.
These two notions are plausible and even widespread (the latter one
is actually the original definition [63]), but they are incompatible
with each other. In the former case, a continuation expects an ex-
pression and returns another expression. In the latter case, a contin-
uation expects a value and returns a final result.

The primary goal of this article is to reconcile these two com-
mon, but contradictory, understandings of continuations as repre-
sentations of the current context and as representations of the rest
of the computation.

1.2 Continuations, authoritatively

When they are mentioned at all, continuations are presented with
considerable variations in textbook and lecture notes. InConcepts
in Programming Languages[47], Mitchell briefly defines a contin-
uation as a function representing the remaining program to eval-
uate; he mentions continuation-passing style as a way to obtain
tail recursion. InProgramming Languages: Theory and Prac-
tice[40], Harper summarily defines a continuation as a control stack
and argues that a formal semantics is much clearer; he mentions
continuation-passing style as a way to “roll one’s own” continua-
tion. In Compiling with Continuations[5], Appel defines a contin-
uation as a function that expresses what to do next; he then makes a
substantial use of continuation-passing style. InEssentials of Pro-
gramming Languages[34], Friedman, Wand, and Haynes define a
continuation as an abstraction of the control context; they dedicate
two chapters to continuation-passing interpreters and transforming
programs into continuation-passing style. InProgramming Lan-
guages and Lambda Calculi[28], Felleisen and Flatt define a con-
tinuation as an inside-out evaluation context in an abstract machine;
they do not consider continuation-passing style. InLisp in Small
Pieces[53], Queinnec defines a continuation as a representation of
all that remains to compute; he mentions contexts as an alternative
representation of continuations.

A secondary goal of this article is to unify these common,
but distinct, representations of continuations. Our thesis is that
Reynolds’s defunctionalization provides the key to this unification,
in the sense that control stacks and evaluation contexts are defunc-
tionalized continuations.



1.3 Prerequisites

We expect a passing familiarity with functional programming (ML),
and we build on the notions of evaluators, abstract machines, CPS
transformation, defunctionalization, and syntactic theories:

Evaluation functions: An evaluator is a compositional function
mapping an abstract-syntax tree to an expressible value, if
there is one; it implements a denotational semantics [58].

Abstract machines: An abstract machine is a transition function
over computational states; it implements an operational se-
mantics [52].

CPS transformation: A program is transformed into
continuation-passing style (CPS) by naming all of its
intermediate results, sequentializing their computation, and
introducing continuations. Each CPS transformation encodes
an evaluation order [20, 41, 51, 57, 62].

Defunctionalization: A program is defunctionalized by replacing
each of its function spaces by a first-order data type and a first-
order apply function [56]. Each data type enumerates all the
function abstractions that may give rise to inhabitants of the
corresponding function space [7, 8, 13, 21, 49, 56, 65].

A particular case of defunctionalization is closure conversion:
in an evaluator, closure conversion amounts to replacing each
of the function spaces in expressible and denotable values by a
tuple, and inlining the corresponding apply function [46, 56].
(Other styles of closure conversion exist, though [6].)

Syntactic theories: A syntactic theory provides a reduction re-
lation on expressions by defining syntax, values, evaluation
contexts, and redexes [26, 28, 70]. For example, a syntactic
theory for arithmetic expressions is specified as follows.

Syntax:e ::= n | e+e

Values:n

Redexes:n+n′

Evaluation contexts:E ::= [ ] | E[n+[ ]] | E[[ ]+e]

Plugging an expressione into a contextE:

plug([ ],e) = e
plug(E[n+[ ]],e) = plug(E,n+e)
plug(E[[ ]+e′],e) = plug(E,e+e′)

Reduction relation:E[n+n′]→ E[n′′], wheren′′ is the sum of
n andn′.

These definitions satisfy a “unique decomposition” lemma
[70]: any expressione that is not a value can be uniquely de-
composed into an evaluation contextE and a redexn+n′ such
thate= plug(E,n+n′).

From syntactic theory to abstract machine: Nielsen and the au-
thor have established the conditions under which one can de-
forest an evaluation function when it is defined as the transi-
tive closure of one-step reduction in a syntactic theory [22].
At each step, a term is decomposed into an evaluation con-
text and a redex, the redex is contracted, and the contractum
is plugged into the evaluation context. Deforesting such an
evaluation function makes it possible to avoid the construc-
tion of intermediate expressions. Our key point is to con-
struct a “refocus” function that makes it possible to replace the
decompose-contract-plug-decompose-contract-plug-... loop
by an initial decomposition followed by a contract-refocus-
contract-refocus-... loop. The result is an abstract machine.

For example, here is the refocus function corresponding to the
syntactic theory just above:

refocus([ ],n) = n
refocus(E[n′ +[ ]],n) = refocus(E,n′ +n)
refocus(E[[ ]+e],n) = decompose(e,E[n+[ ]])

wheredecomposedecomposes a computation into an evalua-
tion context and a redex.

1.4 Overview
The rest of this article is organized as follows. We first investigate
continuations as evaluation contexts and continuations as the rest
of the computation; to this end, we revisit the simple example of
arithmetic expressions above (Section 2). We then consider theλ-
calculus (Section 3) and analyze further consequences (Section 4).

2 A simple example: arithmetic expressions
To investigate continuations as evaluation contexts and continua-
tions as the rest of the computation, we go through the simple exer-
cise of writing a one-step reduction function and then an evaluator
for arithmetic expressions. We write each of them in direct style,
and we successively CPS-transform them and then defunctionalize
their continuations.

Our arithmetic expressions are minimal: they consist of literals
and additions.

datatype exp = VALUE of value
| COMP of comp

and value = LIT of int
and comp = ADD of exp * exp

Literals are the only values and additions are the only computations.

2.1 A one-step reduction function
We write the one-step reduction function by recursive descent, us-
ing the recursive calls to reach the left-most-innermost redex, and
constructing the reduced expression at return time:

(* reduce1 : comp -> exp *)
fun reduce1 (ADD (VALUE (LIT n1), VALUE (LIT n2)))

= VALUE (LIT (n1 + n2))
| reduce1 (ADD (VALUE v1, COMP c2))
= COMP (ADD (VALUE v1, reduce1 c2))

| reduce1 (ADD (COMP c1, e2))
= COMP (ADD (reduce1 c1, e2))

We then CPS-transformreduce1:

(* reduce1c : comp * (exp -> ’a) -> ’a *)
fun reduce1c (ADD (VALUE (LIT n1), VALUE (LIT n2)), k)

= k (VALUE (LIT (n1 + n2)))
| reduce1c (ADD (VALUE v1, COMP c2), k)
= reduce1c (c2, fn e2 => k (COMP (ADD (VALUE v1, e2))))

| reduce1c (ADD (COMP c1, e2), k)
= reduce1c (c1, fn e1 => k (COMP (ADD (e1, e2))))

Finally, we defunctionalize the continuations inreduce1c. We as-
sume an initial continuation that is the identity function, and there-
fore the polymorphic type variable in the type ofreduce1c is spe-
cialized toexp. Three functional abstractions can build inhabitants
in the function spaceexp -> exp. The first is the initial continua-
tion and it has no free variables. The second is the continuation in
the second clause, and it hasv1 andk as free variables. The third
is the continuation in the third clause, and it hase2 andk as free
variables. The data type of defunctionalized continuations has thus
three constructors.



datatype cont = CONT0
| CONT1 of value * cont
| CONT2 of exp * cont

(* apply : cont * exp -> exp *)
fun apply (CONT0, e)

= e
| apply (CONT1 (v1, k), e2)
= apply (k, COMP (ADD (VALUE v1, e2)))

| apply (CONT2 (e2, k), e1)
= apply (k, COMP (ADD (e1, e2)))

(* reduce1cd : comp * cont -> exp *)
fun reduce1cd (ADD (VALUE (LIT n1), VALUE (LIT n2)), k)

= apply (k, VALUE (LIT (n1 + n2)))
| reduce1cd (ADD (VALUE v1, COMP c2), k)
= reduce1cd (c2, CONT1 (v1, k))

| reduce1cd (ADD (COMP c1, e2), k)
= reduce1cd (c1, CONT2 (e2, k))

We observe that the data typecont is isomorphic to the data type
of evaluation contexts for arithmetic expressions, and that its apply
function coincides with the corresponding plug function.Evalu-
ation contexts, together with their plug function, are therefore a
representation of the continuation of a one-step reduction function.

2.2 An evaluation function

We write an evaluation function by recursive descent:

(* eval : exp -> int *)
fun eval (VALUE (LIT n))

= n
| eval (COMP (ADD (e1, e2)))
= (eval e1) + (eval e2)

(* main : exp -> int *)
fun main e

= eval e

We then CPS-transformeval:

(* evalc : exp * (int -> ’a) -> ’a *)
fun evalc (VALUE (LIT n), k)

= k n
| evalc (COMP (ADD (e1, e2)), k)
= evalc (e1,

fn n1 => evalc (e2,
fn n2 => k (n1 + n2)))

(* main : exp -> int *)
fun main e

= eval (e, fn n => n)

Finally, we defunctionalize the continuations inevalc. The initial
continuation is the identity function and therefore the polymorphic
type variable in the type ofevalc is specialized toint. Three func-
tional abstractions can build inhabitants in the function spaceint
-> int. The first is the initial continuation and it has no free vari-
ables. The second is the inner continuation in theADD clause, and it
hasn1 andk as free variables. The third is the outer continuation in
theADD clause, and it hase2 andk as free variables. The data type
of defunctionalized continuations thus has three constructors. Due
to the recursive call toevalc in the outer continuation, the apply
function of defunctionalized continuations and the defunctionalized
version ofevalc are mutually recursive:

datatype cont = CONT0
| CONT1 of int * cont
| CONT2 of exp * cont

(* apply : cont * int -> int *)
fun apply (CONT0, n)

= n
| apply (CONT1 (n1, k), n2)
= apply (k, n1 + n2)

| apply (CONT2 (e2, k), n1)
= evalcd (e2, CONT1 (n1, k))

(* evalcd : exp * cont -> int *)
and evalcd (VALUE (LIT n), k)

= apply (k, n)
| evalcd (COMP (ADD (e1, e2)), k)
= evalcd (e1, CONT2 (e2, k))

(* main : exp -> int *)
fun main e

= eval (e, CONT0)

We observe that the data typecont is isomorphic to the data type
of evaluation contexts for arithmetic expressions, and that its apply
function coincides with the corresponding refocus function.Evalu-
ation contexts, together with their refocus function, are therefore a
representation of the continuation of an evaluation function.

2.3 Conclusion
Continuations have two sides: they can represent the context for
one-step reduction and they can represent the rest of the computa-
tion for evaluation. Common to both sides is the notion of evalua-
tion context:

evaluation
contexts

��
��

��

��>
>>

>>
>>

>>
>

one-step
reduction

evaluation

• Evaluation contexts, together with the plug interpretation, are
the defunctionalized representation of the continuation of a
one-step reducer.

• Evaluation contexts, together with the refocus interpretation,
are the defunctionalized representation of the continuation of
an evaluator.

Identifying these two representations of evaluation contexts makes
it possible to reconcile the two common—but contradictory—
understandings of continuations as representations of the current
context and as representations of the rest of the computation.

Evaluation contexts were first proposed in Felleisen’s PhD the-
sis [26] and since then they have had a clear impact in the formal
study of programming languages. Yet they have never before been
formally connected with the continuation of a one-step reduction
function or with the continuation of an evaluation function.

It takes some skill to define evaluation contexts. Until the
unique-decomposition lemma is proven, one is never sure whether
the enumeration is complete and whether it is not somehow redun-
dant. In contrast, the characterization of evaluation contexts as a de-
functionalized continuation in a recursive descent to locate the next
redex provides both a guideline and a security. Also, the unique-
decomposition lemma holds as a corollary when one starts from a
compositional recursive descent.



structure Eval0
= struct

datatype expval = FUNCT of denval -> expval
withtype denval = expval

(* eval : term * denval list -> expval *)
fun eval (IND n, e)

= List.nth (e, n)
| eval (ABS t, e)

= FUNCT (fn v => eval (t, v :: e))
| eval (APP (t0, t1), e)

= let val (FUNCT f) = eval (t0, e)
in f (eval (t1, e))
end

(* main : term -> expval *)
fun main t

= eval (t, nil)
end

Figure 1. Canonical evaluation-order dependent evaluator

3 A constructive corollary of Reynolds’s
evaluation-order dependence

In earlier work, the author and his students have observed that a
defunctionalized CPS program implements an abstract machine [2,
3, 4, 10, 11, 18]. In particular, we have found that Krivine’s ab-
stract machine [15, 39, 45] is the defunctionalized and continuation-
passing counterpart of a closure-converted call-by-value evaluator
for theλ-calculus and that Felleisen et al.’s CEK machine [28, 29,
33] is the defunctionalized and continuation-passing counterpart of
a closure-converted call-by-name evaluator for theλ-calculus [2].

The goal of this section is to show that Krivine’s abstract ma-
chine and the CEK machine can in fact be derived from the same
evaluator for theλ-calculus. This evaluator is the most canonical
one for theλ-calculus: it is in direct style, higher-order, composi-
tional, and with an environment. As pointed by Reynolds [56], it
is also evaluation-order dependent: if the evaluation order of the
defining language is call by name (resp. call by value), the evalu-
ation order of the defined language is also call by name (resp. call
by value). We specify this evaluation order with the corresponding
CPS transformation:

• Our implementation of the abstract syntax of theλ-calculus is
as follows:

datatype term = IND of int (* de Bruijn index *)
| ABS of term
| APP of term * term

Variables are represented by their lexical offset (i.e., their de
Bruijn index).

• Figure 1 displays an evaluation-order dependent evaluator in
the concrete syntax of Standard ML. This evaluator is compo-
sitional (all recursive calls on the right side of the equal sign
are made over proper sub-parts of the terms on the left side)
and higher order (the domain of expressible values is a func-
tion space), with an environment (a list of denotable values).

• Figure 2 displays a first-order counterpart of the evaluator of
Figure 1, again in the syntax of Standard ML. This evaluator
was obtained by in-place defunctionalization of the express-
ible values, i.e., closure conversion [46, 56].

structure Eval1
= struct

datatype expval = FUNCT of term * denval list
withtype denval = expval

(* eval : term * denval list -> expval *)
fun eval (IND n, e)

= List.nth (e, n)
| eval (ABS t, e)

= FUNCT (t, e)
| eval (APP (t0, t1), e)

= let val (FUNCT (t’, e’)) = eval (t0, e)
in eval (t’, (eval (t1, e)) :: e’)
end

(* main : term -> expval *)
fun main t

= eval (t, nil)
end

Figure 2. Evaluator of Figure 1, closure-converted

structure Eval1n
= struct

datatype expval = FUNCT of term * denval list
withtype denval = (expval -> expval) -> expval

(* eval : term * denval list * (expval -> expval) *)
(* -> expval *)
fun eval (IND n, e, k)

= List.nth (e, n) k
| eval (ABS t, e, k)
= k (FUNCT (t, e))

| eval (APP (t0, t1), e, k)
= eval (t0, e, fn (FUNCT (t’, e’)) =>

eval (t’, (fn k’ => eval (t1, e, k’)) :: e’, k))

(* main : term -> expval *)
fun main t

= eval (t, nil, fn v => v)
end

Figure 3. Call-by-name CPS counterpart of Figure 2

structure Eval1v
= struct

datatype expval = FUNCT of term * denval list
withtype denval = expval

(* eval : term * denval list * (expval -> expval) *)
(* -> expval *)
fun eval (IND n, e, k)

= k (List.nth (e, n))
| eval (ABS t, e, k)
= k (FUNCT (t, e))

| eval (APP (t0, t1), e, k)
= eval (t0, e, fn (FUNCT (t’, e’)) =>

eval (t1, e, fn v1 =>
eval (t’, v1 :: e’, k)))

(* main : term -> expval *)
fun main t

= eval (t, nil, fn v => v)
end

Figure 4. Call-by-value CPS counterpart of Figure 2



structure Eval1nd
= struct

datatype expval = FUNCT of term * denval list
and denval = THUNK of term * denval list

datatype cont = CONT0
| CONT1 of term * denval list * cont

(* eval : term * denval list * cont -> expval *)
fun eval (IND n, e, k)

= let val (THUNK (t’, e’)) = List.nth (e, n)
in eval (t’, e’, k)
end

| eval (ABS t’, e’, CONT1 (t1, e, k))
= eval (t’, (THUNK (t1, e)) :: e’, k)

| eval (APP (t0, t1), e, k)
= eval (t0, e, CONT1 (t1, e, k))

| eval (ABS t, e, CONT0)
= FUNCT (t, e)

(* main : term -> expval *)
fun main t

= eval (t, nil, CONT0)
end

Figure 5. Defunctionalized counterpart of Figure 3

– Source syntax: t ::= n | λt | t0 t1

– Expressible values (closures):v ::= [t, e]

– Initial transition, transition rules, and final transition:

t ⇒ 〈t, nil, nil〉
〈n, e, s〉 ⇒ 〈t ′, e′, s〉

wherenth(e, n) = [t ′, e′]
〈λt ′, e′, [t1, e] :: s〉 ⇒ 〈t ′, [t1, e] :: e′, s〉

〈t0 t1, e, s〉 ⇒ 〈t0, e, [t1, e] :: s〉
〈λt, e, nil〉 ⇒ [t, e]

The abstract machine operates on triples consisting
of a term, an environment, and a stack of expressible
values.

Each line in the table matches a clause in Figure 5.

Figure 6. Krivine’s abstract machine

structure Eval1vd
= struct

datatype expval = FUNCT of term * denval list
withtype denval = expval

datatype cont = CONT0
| CONT1 of denval * cont
| CONT2 of term * denval list * cont

(* eval : term * denval list * cont -> expval *)
fun eval (IND n, e, k)

= apply (k, List.nth (e, n))
| eval (ABS t, e, k)
= apply (k, FUNCT (t, e))

| eval (APP (t0, t1), e, k)
= eval (t0, e, CONT2 (t1, e, k))

and apply (CONT2 (t1, e, k), v0)
= eval (t1, e, CONT1 (v0, k))

| apply (CONT1 (FUNCT (t’, e’), k), v1)
= eval (t’, v1 :: e’, k)

| apply (CONT0, v)
= v

(* main : term -> expval *)
fun main t

= eval (t, nil, CONT0)
end

Figure 7. Defunctionalized counterpart of Figure 4

– Source syntax: t ::= n | λt | t0 t1

– Expressible values (closures):v ::= [t, e]

– Evaluation contexts:

k ::= CONT0 | CONT1(v,k) | CONT2(t,e,k)

– Initial transition, transition rules, and final transition:

t ⇒init 〈t, nil, CONT0〉
〈n, e, k〉 ⇒eval 〈k, v〉

wherenth(e, n) = v
〈λt, e, k〉 ⇒eval 〈k, [t, e]〉

〈t0 t1, e, k〉 ⇒eval 〈t0, e, CONT2(t1,e,k)〉
〈CONT2(t1,e,k), v0〉 ⇒apply 〈t1, e, CONT1(v0,k)〉

〈CONT1([t ′, e′],k), v1〉 ⇒apply 〈t ′, v1 :: e′, k〉
〈CONT0, v〉 ⇒final v

The abstract machine consists of two mutually recur-
sive transition functions. The first transition function
operates on triples consisting of a term, an environ-
ment, and an evaluation context. The second oper-
ates on pairs consisting of an evaluation context and
an expressible value.

Each line in the table matches a clause in Figure 7.

Figure 8. The CEK machine



• Figure 3 displays the call-by-name CPS counterpart of the
evaluator of Figure 2.

• Figure 4 displays the call-by-value CPS counterpart of the
evaluator of Figure 2.

• Figure 5 displays the defunctionalized version of the evaluator
of Figure 3, with the corresponding apply function inlined.
Merging the domains of expressible values and of denotable
values into one (recursive) domain of thunks pairing terms
and environments, and representing the data typecont as a list
yields the transition function of Krivine’s machine (Figure 6).

• Figure 7 displays the defunctionalized version of the evaluator
of Figure 4. It corresponds to the transition function of the
CEK machine (Figure 8).

Therefore, if the CPS transformation is call-by-name, the resulting
transition function is that of Krivine’s abstract machine, and if the
CPS transformation is call-by-value, the resulting transition func-
tion is that of the CEK machine:

canonical evaluator
(Figure 1)

closure conv.

��
Figure 2

call-by-name
CPS transf.

yysssssssssssssss call-by-value
CPS transf.

%%KKKKKKKKKKKKKKK

Figure 3

defunct.

��

Figure 4

defunct.

��
Krivine’s machine
(Figures 5 and 6)

CEK machine
(Figures 7 and 8)

Reynolds’s point was that in general the evaluation order of the
defining language, in a definitional interpreter, determines the eval-
uation order of the defined language if the definitional interpreter is
in direct style (and does not use thunks). The author and his stu-
dents have recently shown that a call-by-name interpreter leads one
to Krivine’s abstract machine and that a call-by-value interpreter
leads one to the CEK machine [2]. It is a further (and new) conse-
quence of the embodiment of evaluation order in a CPS transforma-
tion [41] that Krivine’s abstract machine and the CEK machine can
in fact be derived from thesamecanonical evaluator. In particular,
other CPS transformations would lead to other abstract machines.

Krivine’s abstract machine has been discovered, the CEK ma-
chine has been invented, and each of them has been celebrated in-
dependently and on its own right. Yet, as shown here, they are two
sides of the same coin.

4 Consequences

We review further consequences of the connection between evalua-
tion contexts, continuations, and the rest of the computation.

4.1 Designing syntactic theories and abstract
machines

Beside making it simple to connect one-step reducers and evalua-
tors, the interpretation of evaluation contexts with a plug function
or with a refocus function has direct consequences for designing
syntactic theories and abstract machines:

• For programming languages where one can write a one-
step reducer using recursive descent, one can mechanically
construct the grammar of evaluation contexts and the cor-
responding plug function, and rest assured that the unique-
decomposition lemma holds [70].

Furthermore, given such a one-step reduction machinery, one
can mechanically construct the corresponding abstract ma-
chine [22].

• For programming languages where one can write an evaluator
using recursive descent, one can mechanically construct the
grammar of evaluation contexts and the corresponding refocus
function. The result is the transition function of an abstract
machine.

Conversely, one can see abstract machines such as Kriv-
ine’s machine and the CEK machine as defunctionalized
continuation-passing interpreters.

The two points above are not just an academic observation—they
have concrete consequences in that they have made it possible for
the author and his students to uniformly transform a given evaluator
into an abstract machine that was independently invented or discov-
ered, to uniformly exhibit the evaluator underlying a given abstract
machine, and to design new evaluators, new abstract machines, and
new virtual machines [1, 2]. Beside Krivine’s machine and the
CEK machine, examples include Landin’s SECD machine, Han-
nan and Miller’s CLS machine, Curien et al.’s Categorical Abstract
Machine, Schmidt’s VEC machine, and Leroy’s Zinc machine as
well as abstract machines for non-strict functional languages [3],
logic-programming languages [11], functional languages with com-
putational effects, including the security technique of stack inspec-
tion [4], imperative languages, and object-oriented languages. In
clear contrast, such evaluators and machines are usually considered
independently and on a case-by-case basis. And when abstract ma-
chines are derived, it is the medium (i.e., the derivation) rather than
the result that tends to be the message [66, 67].

In particular, starting from a monad-based evaluator for the
lambda-calculus, we can pick an arbitrary monad and mechanically
construct an evaluator, an abstract machine, and a syntactic theory
for the corresponding computational effect. In striking contrast,
abstract machines and syntactic theories for computational effects
have been designed in isolation and reported as such in the litera-
ture.

On the other hand, syntactic theories have also been successfully
used in situations where the unique-decomposition lemma does not
hold, e.g., Concurrent ML [55]. Such situations would require first-
class continuations, which are out of scope here.



4.2 Normalization

Another application of the the insight presented in Section 3 and of
the derivation reported at PPDP 2003 [2] concerns (not necessarily
type-directed) normalization functions as encountered in the area
of normalization by evaluation [9, 16, 25]. The author and his stu-
dents have derived abstract machines as well as virtual machines for
normalization [1]. Specifically, we have shown that a call-by-name
normalization function yields a machine that generalizes Krivine’s
machine, and that a call-by-value normalization function yields a
machine that generalizes the CEK machine. In the light of Sec-
tion 3, though, the author now realizes that these two machines are
in fact derived from thesamenormalization function.

In noticeable contrast, existing machines for normalization have
been designed in isolation rather than by derivation [15, 36].

4.3 Delimited continuations

In CPS, all calls are tail calls. Yet in some situations, it is very con-
venient to re-initialize a continuation and to mix CPS with non-tail
calls. In a program that re-initializes continuations and where not
all calls are tail calls, a continuation no longer represents the rest
of the computation. Instead, it is delimited by the re-initialization.
Capturing such a continuation yields a first-class continuation that
returns to its point of activation. Such first-class continuations can
be composed. (In contrast, first-class continuations obtained by
call/cc-like control operators do not return to their point of acti-
vation and therefore they cannot be composed.)

Fifteen years ago, Felleisen introduced an operator to delimit
control (a “prompt”) together with other operators to abstract de-
limited control [27]. These control operators were specified using a
representation of control as a list of activation records. Delimiting
control amounted to putting a mark on this list, abstracting delim-
ited control amounted to making a copy of the list up to the closest
mark, and activating a delimited continuation amounted to concate-
nating the copied list to the current list of activation records [30].
Felleisen’s work triggered a series of alternative control operators,
all based on representing control as a list of activation records in-
terspersed with control marks [38, 42, 43, 48, 54, 59, 60].

To the author, Felleisen’s operator for delimiting control fit-
ted precisely a pervasive pattern of functional programming with
layered continuations, together with another control operator,
shift [20]. Consequently, the two control operators to delimit and
to abstract control enjoy a number of applications—in fact, they
correspond to computational monads [31]—and they are still the
topic of study today [35, 44]. Furthermore, they generalize directly
to the CPS hierarchy [10, 19, 23], which also corresponds to layered
monads [32].

These two lines of work have been opposed because one repre-
sents control as a list of activation records, as in an initial algebra,
and the other as a continuation function, as in a final algebra [30].
This opposition continues today when control is only considered
as a list of activation records, fit for arbitrary surgery.1 The two
representations, however, could be synergized, e.g., by seeing the

1The danger of this surgery is that it is so plausible. For exam-
ple, in the first implementation of Lisp, it was sweepingly plausible
to push the bindings of the formals and the actuals on the stack at
call time, and to pop them off at return time. The result was dy-
namic scope.

former as a defunctionalized version of the latter and by identify-
ing when the latter is not a functional version of the former. For
example, Felleisen’sF + control operator appears to have no CPS
counterpart [20, Section 5.3].

Another advantage of characterizing delimited continuations us-
ing repeated CPS transformations is that, through the derivation
outlined in Section 3 (closure conversion, CPS transformations
(note the plural), and defunctionalization), one obtains abstract ma-
chines for delimited control [17]. In these machines, delimited con-
trol is represented through a series of control stacks, one for each
layered continuation.2

4.4 Landin’s SECD machine

Imagine an environment-based, call-by-value evaluator for theλ-
calculus with a callee-save strategy, that furthermore delimits con-
trol when evaluating the body of aλ-abstraction. This evaluator
operates on the same representation ofλ-terms as in Section 3.

datatype value = FUN of value -> value

(* eval : term * value list -> value * value list *)
fun eval (IND n, e)

= (List.nth (e, n), e)
| eval (ABS t, e)

= (FUN (fn v => reset (fn () => #1 (eval (t, v :: e)))),
e)

| eval (APP (t0, t1), e)
= let val (v1, e) = eval (t1, e)

val (v0, e) = eval (t0, e)
in apply (v0, v1, e)
end

(* apply : value * value * value list ->
value * value list *)

and apply (FUN f, v, e)
= (f v, e)

(* evaluate : term -> value *)
fun evaluate t

= reset (fn () => #1 (eval (t, nil)))

From this evaluator, one can reconstruct Landin’s SECD machine
as follows:

1. closure conversion of the function space in the domain of val-
ues;

datatype value = FUN of term * E
withtype E = value list

2. introduction of a data stack to hold the intermediate results of
eval;

eval : term * S * E -> S * E
withtype S = value list

and E = value list

2The author wishes to emphasize this point with an anecdote
about Gasbichler and Sperber’s implementation of delimited con-
tinuations in Scheme 48 [35]. In the course of their work, Gas-
bichler and Sperber consulted each of the authors of control oper-
ators for delimited control, to make sure that their implementation
of each delimited-control operator was accurate. This consultation
apparently took some time to stabilize. In sharp contrast, it reduced
to one e-mail reply from the author, with the guideline of checking
that the CPS counterpart of the implementation matches the CPS
specification of shift and reset. The next time the author heard of
Gasbichler and Sperber’s work, it was in the list of accepted papers
at ICFP 2002.



3. CPS transformation;
eval : term * S * E * C -> value
withtype S = value list

and E = value list
and C = S * E -> value

4. second CPS transformation, to get rid of the non-tail call due
to the presence ofreset;

eval : term * S * E * C * D -> ’a
withtype S = value list

and E = value list
and C = S * E * D -> ’a
and D = value -> ’a

5. defunctionalization of the two layered continuations;

6. fusion of the resulting mutually recursive functions into one.

The SECD machine is a transition function operating on a four-
component state: a stack register, an environment register, a control
register, and a dump register [46]. The stack register holds the data
stack introduced above; the environment register holds the environ-
ment threaded in the evaluator above; the control register holds the
first continuation in defunctionalized form; and the dump register
holds the second continuation in defunctionalized form. We there-
fore claim that the denotational essence of the SECD machine is
this evaluator, with its callee-save strategy for the environment and
its control delimiter. The rest—stack register, control register, and
dump register—are mere operational artifacts.

This derivation is documented in a BRICS technical report [18].
It is based on the insight of Section 2 and at the origin of the deriva-
tion of Section 3. It solves a long-standing open problem about
the particular architecture of the SECD machine, which had never
been fully explained—though many variations and simplifications
exist. These variations and simplifications (as well as arbitrary new
ones) can be obtained by tuning this evaluator and then transform-
ing it into an abstract machine. For example, omitting the control
delimiter (which operationally is unused) yields an SEC machine.

5 Conclusion and current work

We have reconciled the notion of continuations as evaluation con-
texts with the notion of continuations as representations of the rest
of the computation. To this end, we have factored the continuation
of a single-step reducer and the continuation of an evaluator as the
same evaluation contexts with two different interpretations:

evaluation

plug

��
��

��

��>
>>

>>
>>

>>
>

one-step
reduction

evaluation

As a consequence of this factorization, we have shown that the
two best-known abstract machines for theλ-calculus can be derived
from the same canonical evaluator for theλ-calculus:

canonical evaluator

��
Krivine’s machine CEK machine

This derivation provides a constructive corollary of Reynolds’s
historical warning about the evaluation order of defining lan-
guages [56] and it scales to normalization functions and abstract
machines for normalization. It is an instance of a functional corre-
spondence that lets one reconstruct known abstract machines, con-
struct new ones, e.g., with monadic computational effects, system-
atically equip them with stack inspection [14], and mechanically
construct the corresponding syntactic theories.

In this article, we have considered continuations in the opera-
tional setting of reduction, evaluation, and normalization. They are,
however, ubiquitous in many other areas, such as semantics [64]
and logic [37] as well as in operating-systems services [24, 69].

